package body Discrete_Set is



-- Function determines whether a single element is in a given set

-- and returns a boolean true if the element is in the set and false

-- if the element is not in the set
   ----------------------------------------------------------------------------
   function Is_Member (Set     : in Set_Type;  
                       Element : in Element_Type) return Boolean is  
   begin
      return Set(Element);
   end Is_Member;



-- Returns both of the sets left and right combined together to form

-- one set. This function adds two sets together. (Disjunction)
   ----------------------------------------------------------------------------
   function "+" (Left : in Set_Type;  Right : in Set_Type) return Set_Type is
   begin
      return  Left or Right;
   end "+";


-- Function adds an element from 'Right' to the set 'Left' by setting

-- its boolean value to true.
   ----------------------------------------------------------------------------
   function "+" (Left  : in Set_Type; 
                 Right : in Element_Type) return Set_Type is
      Result : Set_Type;
   begin
      Result := Left;
      Result(Right) := True;   -- Add the new element to the set
      return Result;
   end "+";


-- Function adds an element from 'Left' to the set 'Right' by setting

-- its boolean value to true.
   ----------------------------------------------------------------------------
   function "+" (Left  : in Element_Type;
                 Right : in Set_Type) return Set_Type is
      Result : Set_Type;
   begin
      Result := Right;
      Result(Left) := True;   -- Add the new element to the set
      return Result;
   end "+";


-- Returns a set with any elements contained in both sets 'Left' and 

-- 'Right'. (Conjunction)
   ----------------------------------------------------------------------------
   function "*" (Left : in Set_Type;  Right : in Set_Type) return Set_Type is
   begin
      return  Left and Right;
   end "*";


-- Function deletes the set 'Right' by setting its value to false when it 

-- returns the 'and not Right' statement.  Left will remain true, thus

-- not deleted.
   ----------------------------------------------------------------------------
   function "-" (Left : in Set_Type;  Right : in Set_Type) return Set_Type is
   begin
      return Left and not Right;
   end "-";


-- Function will remove a certain element from a set by setting its boolean

-- value to false.
   ----------------------------------------------------------------------------
   function "-" (Left  : in Set_Type;
                 Right : in Element_Type) return Set_Type is
      Result : Set_Type;
   begin
      Result := Left;
      Result(Right) := False;   -- Remove the element from the set
      return Result;
   end "-";


-- Function compares whether two sets are identical or if one set has more

-- elements in it. If the (Left - Right) statement is equal to Empty_Set

-- then the boolean Is_A_Subset will be true proving that 'Right' is not

-- a Subset of 'Left'.  If Is_A_Subset is false then that means 'Left' has 

-- additional element(s) and 'Right' is a subset.
   ----------------------------------------------------------------------------
   function "<=" (Left : in Set_Type; Right : in Set_Type) return Boolean is
      Is_A_Subset : Boolean;
   begin
      Is_A_Subset := (Left - Right) = Empty_Set;
      return Is_A_Subset;
   end "<=";


-- Function determines whether 'Left' is a proper subset of 'Right'.  If the

-- boolean variable returns false then 'Left' is not a proper subset of 'Right'

-- and will return true otherwise.
   ----------------------------------------------------------------------------
   function "<" (Left : in Set_Type; Right : in Set_Type) return Boolean is
      Result : Boolean;
   begin
      if Left = Right then  -- If the sets are equal, not a proper subset
         Result := False;
      else
         Result := Left <= Right;  -- If not equal, test for subset
      end if;                      -- using the function above
      return Result;
   end "<";


-- Function compares whether two sets are identical or if one set has more

-- elements in it. If 'Right' equals left then it is not a subset of 'Left'

-- and the two sets are equal. If 'Right' is less than 'Left' it is a subset 

-- of 'Left'. Either of the previous mentioned conditions will return a true

-- boolean value. If it returns false then 'Right' has more elements in it than

-- 'Left'
   ----------------------------------------------------------------------------
   function ">=" (Left : in Set_Type; Right : in Set_Type) return Boolean is
   begin
      return  Right <= Left;   -- Reverse the order of the 
                               -- parameters and use the operation 
                               -- we've already written
   end ">=";


-- Function determines if 'Right' is a proper subset of 'Left' and returns

-- a true boolean value if this is correct.
   ----------------------------------------------------------------------------
   function ">" (Left : in Set_Type; Right : in Set_Type) return Boolean is
   begin
      return  Right < Left;     -- Reverse the order of the 
                                -- parameters and use the operation 
                                -- we've already written
   end ">";

end Discrete_Set;
 
